finite subsets of unions of chains

Johan G. F. Belinfante
2008 January 4

```
In[1]:= SetDirectory["l:"]; << goedel.08jan04a; << tools.m

        :Package Title: goedel.08jan04a           2008 January 4 at 4:20 p.m.

        It is now:  2008 Jan 4 at 19:29

        Loading Simplification Rules

        TOOLS.M                                   Revised 2008 January 2

        weightlimit = 40
```

summary

Consider for example the problem of showing that the union of a chain of subgroups is a subgroup. The conditions that need to be checked only involve a finite number of points, and all these points belong to one of the members of this chain. It is this intuitively clear fact that is formally proved in this notebook. In general, if a non-empty finite set is covered by a nest of sets, then one of the members of the nest contains all of them. One might expect that some form of induction would be needed to prove this result, or a theorem about finite choices; not the axiom of choice, of course, since only a finite number of points are involved. The derivation to be presented does not explicitly use induction, nor finite choice functions, but instead uses the theorem that a nonempty finite chain of sets has a largest member, and the idea that if **x** is a chain of sets, then the collection **u** obtained from **x** by restricting to a finite set **t** yields a finite chain. The entire proof involves about a dozen statements, which slightly exceeds what the **GOEDEL** program can deal with comfortably all at once. For this reason two lemmas are used to more or less arbitrarily break up the proof into somewhat smaller pieces.

derivation

Lemma. The collection **u** is either empty or has a largest member.

```
In[2]:= Map[not, SubstTest[and, implies[and[p0, p2], p4], implies[and[p0, p1], p5],
           implies[and[p4, p5], or[p6, p7]], not[implies[and[p0, p1, p2], or[p6, p7]]],
           {p0 → equal[u, image[IMAGE[id[t]], x]], p1 → member[t, FINITE],
            p2 → subclass[P[x], chains[S]], p4 → subclass[P[u], chains[S]],
            p5 -> member[u, FINITE], p6 → empty[u], p7 → member[U[u], u]}]] // Reverse

Out[2]= or[equal[0, u], member[U[u], u], not[equal[u, image[IMAGE[id[t]], x]]],
          not[member[t, FINITE]], not[subclass[cart[x, x], union[S, inverse[S]]]]] == True

In[3]:= (% /. {t → t_, u → u_, x → x_}) /. Equal → SetDelayed
```

Lemma. If the collection **u** has a largest member, then **t** is a subset of some member of **x**.

```
In[4]:= Map[not, SubstTest[and, implies[and[p0, p3], p8],
          implies[and[p0, p7, p8], p9], not[implies[and[p0, p3, p7], p9]],
          {p0 → equal[u, image[IMAGE[id[t]], x]], p3 → subclass[t, U[x]], p7 → member[U[u], u],
           p8 → equal[U[u], t], p9 → member[t, image[inverse[S], x]]}]] // Reverse

Out[4]= or[member[t, image[inverse[S], x]], not[equal[u, image[IMAGE[id[t]], x]]],
         not[member[U[u], u]], not[subclass[t, U[x]]]] == True

In[5]:= (% /. {t → t_, u → u_, x → x_}) /. Equal → SetDelayed
```

Main Theorem. If a finite non-empty set **t** is covered by chain **x** of sets, then **t** is a subset of some member of **x**. Note that the auxiliary variable **u** used in the proof is eliminated from the final statement of the theorem after the proof has been completed.

```
In[6]:= (Map[not, SubstTest[and, implies[and[p0, p1, p2], or[p6, p7]],
           implies[and[p0, p3, p7], p9], implies[and[p0, p3, p6], p10],
           not[implies[and[p0, p1, p2, p3], or[p9, p11]]], implies[and[p3, p10], p11] ,
           {p0 → equal[u, image[IMAGE[id[t]], x]], p1 → member[t, FINITE],
            p2 → subclass[P[x], chains[S]], p3 → subclass[t, U[x]], p6 → empty[u],
            p7 → member[U[u], u], p8 → equal[U[u], t], p9 → member[t, image[inverse[S], x]],
            p10 → empty[x], p11 → empty[t]}]] // Reverse) /. u -> image[IMAGE[id[t]], x]

Out[6]= or[equal[0, t], member[t, image[inverse[S], x]], not[member[t, FINITE]],
         not[subclass[t, U[x]]], not[subclass[cart[x, x], union[S, inverse[S]]]]] == True

In[7]:= or[equal[0, t_], member[t_, image[inverse[S], x_]], not[member[t_, FINITE]],
         not[subclass[t_, U[x_]]], not[subclass[cart[x_, x_], union[S, inverse[S]]]]] := True
```

---

eliminating the variable t

Corollary. This corollary restates the main theorem with one fewer variable. The set variable **t** is eliminated in a standard fashion.

```
In[8]:= Map[equal[V, #] &, SubstTest[class, t,
         implies[and[subclass[P[x], u], member[t, v]], member[t, w]], {u → chains[S],
          v -> intersection[FINITE, P[U[x]]], w -> union[set[0], image[inverse[S], x]]}]]

Out[8]= or[not[subclass[cart[x, x], union[S, inverse[S]]]],
         subclass[intersection[FINITE, P[U[x]]], union[image[inverse[S], x], set[0]]]] == True

In[9]:= or[not[subclass[cart[x_, x_], union[S, inverse[S]]]], subclass[
          intersection[FINITE, P[U[x_]]], union[image[inverse[S], x_], set[0]]]] := True
```

Comment. One could also eliminate the other variable **x** (only for the special case that **x** is a set), but since the result is not particularly pretty, we refrain from doing so.